Oppgave 12

Pris på avkastningsgaranti

In[1]:=
```
<< "PlotLegends`"
```

In[2]:=
```
α = 0.20;
μ = 0.10;
σ = 0.20;
ν = μ - σ^2/2;
δ = 0.05;
γ = 0.03;
n = 100 000;
T = 20;
```

In[10]:=
```
Φ[x_] := CDF[NormalDistribution[0, 1], x];
```

In[11]:=
```
put[s0_, k_, t_] :=
  Module[{d1, d2}, d1 = (Log[s0/k] + δ t)/(σ Sqrt[t]) + (σ Sqrt[t])/2; d2 = d1 - σ Sqrt[t]; k e^(-δ t) Φ[-d2] - s0 Φ[-d1]];
```

$$d1 = \frac{\mathrm{Log}\left[\frac{s0}{k}\right] + \delta\, t}{\sigma\sqrt{t}} + \frac{\sigma\sqrt{t}}{2};\quad d2 = d1 - \sigma\sqrt{t};\quad k\, e^{-\delta t}\,\Phi[-d2] - s0\,\Phi[-d1]$$

In[12]:=
```
p = Which[γ < δ && γ > δ + Log[1 - α],
  x /.FindRoot[x == put[(1 - x) α, e^γ - (1 - x) (1 - α) e^δ, 1], {x, 0}], γ ≤ δ + Log[1 - α],
  Print["Garantien er gratis, fordi bankinnskuddet alene oppfyller garantien"], γ ≥ δ,
  Print["Ingen løsning når garantien er høyere enn den risikofrie avkastningen"]]
```

Out[12]= 0.0117119

In[13]:=
```
f[x_] := x - put[(1 - x) α, e^γ - (1 - x) (1 - α) e^δ, 1];
```

Numerisk metode.

In[14]:=
```
bisection[m_] := Module[{pMin, pMax, pTest}, pMin = 0;
   pMax = 1; pTest = pMax; Do[If[f[pTest] > 0, pMax = pTest, pMin = pTest];
    pTest = (pMin + pMax)/2, {m}]; N[{pMin, pMax}]];
```

In[15]:=
```
bisection[25]
```

Out[15]= {0.0117118, 0.0117119}

Sannsynlighetsfordelinger

In[16]:=
```
Timing[a = Partition[α e^(ν+σ RandomReal[NormalDistribution[0,1],n T]) + (1 - α) e^δ, T];]
```

Out[16]= {0.234, Null}

In[17]:=
```
fCompile = Compile[{{matrise, _Real, 1}}, Fold[(1 + #1) #2 &, 0, matrise]];
```

In[18]:=
```
fgCompile = Compile[{{matrise, _Real, 1}, {γc, _Real}, {pc, _Real}},
   Fold[(1 + #1) Max[e^γc, (1 - pc) * #2] &, 0, matrise]];
```

In[19]:= `Timing[fSimCompile = Table[fCompile[a[[i]]], {i, n}];]`

Out[19]= `{0.219, Null}`

In[20]:= `Timing[fgSimCompile = Table[fgCompile[a[[i]], γ, p], {i, n}];]`

Out[20]= `{0.483, Null}`

In[21]:= $\Psi = 100\left(\frac{\text{fgSimCompile}}{\text{fSimCompile}} - 1\right);$

In[22]:=
```
pdf[data_] := Histogram[data, AspectRatio → 0.3`,
   ImageSize → 500, BaseStyle → {11, FontFamily → "Verdana"}]
```

In[23]:= $N\left[\frac{\text{Length[Select}[\Psi, \#1 < 0.\ \&]]}{n}\right]$

Out[23]= `0.80319`

In[24]:= `pdf[Ψ]`

Out[24]=


Replikerende portefølje

In[25]:=
$h = 250;$
$k = e^{\gamma} - (1 - p)\ (1 - \alpha)\ e^{\delta};$
$s0 = (1 - p)\ \alpha;$

Tid til utløp:

In[28]:= $tT = 1 - \frac{\text{Range}[0, h - 1]}{h};$

In[29]:= $s = \text{FoldList}\left[\#1\ \#2\ \&,\ s0,\ e^{\frac{v}{h} + \frac{\sigma\ \text{RandomReal[NormalDistribution}[0,1],h-1]}{\sqrt{h}}}\right];$

In[30]:= `ListPlot[s, Joined → True]`

Out[30]=


Porteføljen tilpasses etterskuddsvis. På tid t finner jeg den sammensetningen som jeg skulle hatt på tid t-1 for at porteføljen skulle fått samme verdi som opsjonen på tid t. Aksjeandelen er lik den deriverte mhp aksjekursen.

In[31]:= $\text{at} = \text{Table}\left[-\Phi\left[-\left(\frac{\text{Log}\left[\frac{\text{s}[[\text{i}]]}{\text{k}}\right]}{\sigma\sqrt{\text{tT}[[\text{i}]]}} + \frac{\left(\delta + \frac{\sigma^2}{2}\right)\sqrt{\text{tT}[[\text{i}]]}}{\sigma}\right)\right], \{\text{i}, 1, \text{h}\}\right];$

In[32]:= $\text{bt} = \text{Table}\left[(\text{put}[\text{s}[[\text{i}]], \text{k}, \text{tT}[[\text{i}]]] - \text{at}[[\text{i}]]\ \text{s}[[\text{i}]])\ e^{\delta\ \text{tT}[[\text{i}]]}, \{\text{i}, 1, \text{h}\}\right];$

In[33]:= `ListPlot[Table[put[s[[i]], k, tT[[i]]], {i, 1, h}], Joined → True]`

Out[33]=



In[34]:= `farge = {RGBColor[1, 0, 0], RGBColor[0, 1, 0], RGBColor[0, 0, 1]};`

In[35]:= $\text{ListPlot}\left[\frac{\text{s}}{\text{k}}, \text{Joined} \rightarrow \text{True}\right]$

Out[35]=



```
In[36]:= Plot[{s[[Round[t]]], at[[Round[t]]] s[[Round[t]]], bt[[Round[t]]] e^(-δ tT[[Round[t]]])},
  {t, 1, h}, PlotRange → {Automatic, {-0.3`, 0.3`}}, PlotStyle → farge,
  PlotLabel → "Replikerende  portefølje", Frame → True,
  PlotLegend → {" Aksjekurs", " Aksjer  i  replikerende  portefølje",
    " Obligasjoner  i  replikerende  portefølje"}, LegendTextSpace → 20,
  LegendSize → 1.46`, LegendPosition → {-0.665`, -0.6`}, ImageSize → 500,
  AspectRatio → 0.3`, LegendBackground → GrayLevel[0.9`],
  LegendShadow → {0, 0}, BaseStyle → {11, FontFamily → "Verdana"}]
```

Out[36]=




```
Plot[{s[[Round[t]]], (1 + at[[Round[t]]]) s[[Round[t]]],
   (1 - p) (1 - α) e^(δ (1-tT[[Round[t]]])) + bt[[Round[t]]] e^(-δ tT[[Round[t]]])}, {t, 1, h},
  PlotRange → {Automatic, {0, 1}}, PlotStyle → farge, PlotLabel → "Total  portefølje",
  Frame → True, PlotLegend → {" Aksjekurs", " Aksjer  i  total  portefølje",
    " Obligasjoner  i  total  portefølje"}, LegendTextSpace → 20,
  LegendSize → 1.46`, LegendPosition → {-0.665`, -0.6`}, ImageSize → 500,
  AspectRatio → 0.3`, LegendBackground → GrayLevel[0.9`],
  LegendShadow → {0, 0}, BaseStyle → {11, FontFamily → "Verdana"}]
```



In[37]:=
```
totalPorteføljeMedGaranti = (1 + at) s + (1 - p) (1 - α) ⅇ^(δ (1-tT)) + bt ⅇ^(-δ tT);
totalPorteføljeUtenGaranti = (α s)/s0 + (1 - α) ⅇ^(δ (1-tT));
Print["Kontoverdier: ", Last[totalPorteføljeMedGaranti],
  " med garanti, og ", Last[totalPorteføljeUtenGaranti], " uten garanti."]
Print["Her bør vi få 1 hvis garantien er effektiv: ",
  1/ⅇ^γ Take[totalPorteføljeMedGaranti ⅇ^(δ tT), -5]]
Print["Her bør vi få 1 hvis garantien ikke er effektiv: ",
  Last[(totalPorteføljeUtenGaranti (1 - p)) / totalPorteføljeMedGaranti]]
Print["Her bør vi generelt få 1: ", Take[(totalPorteføljeMedGaranti ⅇ^(δ tT)) /
    (Max[ⅇ^γ, #1] &) /@ (totalPorteføljeUtenGaranti (1 - p) ⅇ^(δ tT)), -5]]
```

```
Kontoverdier: 1.15647 med garanti, og 1.17018 uten garanti.

Her bør vi få 1 hvis garantien er effektiv:
 {1.11766, 1.11979, 1.11994, 1.12391, 1.12252}

Her bør vi få 1 hvis garantien ikke er effektiv: 1.

Her bør vi generelt få 1: {1., 1., 1., 1., 1.}
```

Manglende selvfinansiering:

In[43]:=
```
harPåTidt = Table[at[[i - 1]] s[[i]] + bt[[i - 1]] ⅇ^(-δ tT[[i]]), {i, 2, h}];
skalHaPåTidt = Table[at[[i]] s[[i]] + bt[[i]] ⅇ^(-δ tT[[i]]), {i, 2, h}];
tilskudd = skalHaPåTidt - harPåTidt;
Min[tilskudd]
Max[tilskudd]
Mean[tilskudd]
```

Out[46]= -0.0000419219

Out[47]= 0.000169014

Out[48]= 1.27843×10^{-9}

In[49]:=
```
ListPlot[tilskudd, Frame → True,
 PlotStyle → RGBColor[1, 0, 0], PlotRange → All, PlotLabel → "Tilskudd",
 FrameTicks → {{{1, "0"}, {h, ToString[h]}}, Table[0.005` i, {i, -5, 5}], None, None},
 ImageSize → 500, AspectRatio → 0.3`, BaseStyle → {11, FontFamily → "Verdana"}]
```

Out[49]=



Kontantverdi av tilskudd som andel av prisen på garantien:

In[50]:=
```
beregnTilskudd := Module[{}, s = FoldList[#1 #2 &, s0, ⅇ^(ν/h + σ RandomReal[NormalDistribution[0,1],h-1]/√h)];
    at = Table[-Φ[-(Log[s⟦i⟧/k]/(σ √tT⟦i⟧) + ((δ + σ²/2) √tT⟦i⟧)/σ)], {i, 1, h}];
    bt = Table[(put[s⟦i⟧, k, tT⟦i⟧] - at⟦i⟧ s⟦i⟧) ⅇ^(δ tT⟦i⟧), {i, 1, h}];
    skalHaPåTidt = Table[at⟦i⟧ s⟦i⟧ + bt⟦i⟧ ⅇ^(-δ tT⟦i⟧), {i, 2, h}];
    harPåTidt = Delete[at, -1] Delete[s, 1] + Delete[bt, -1] Delete[ⅇ^(-δ tT), 1];
    tilskudd = skalHaPåTidt - harPåTidt; (Delete[ⅇ^(-δ Reverse[tT]), -1].tilskudd)/p];
```

In[53]:=
```
Timing[replTilskudd = Table[beregnTilskudd, {1000}];]
```

Out[53]= {82.182, Null}

In[54]:=
```
replTilskudd // Histogram
```

Out[54]=



In[55]:=
```
replTilskudd // Mean
```

Out[55]= -0.004759